Professor: Loren Tauer
TA: Mengwei Lin

Lab Session 2

AEM 4110
Sep 13, 2019

Elasticity

The elasticity, or the percentage change in y given a 1% change in x, is:

$$\frac{\triangle\%y}{\triangle\%x} = \frac{\frac{dy}{y}}{\frac{dx}{x}} = \frac{dy}{dx}\frac{x}{y} = \frac{dlny}{dlnx}$$

Models

1. Log-log Model: $ln(y) = \beta_1 + \beta_2 ln(x)$
   Take derivative with respect to x on both sides (t.d.wrt.x): $\frac{dln(y)}{dx} = \beta_2 \frac{dln(x)}{dx}$
   Elasticity: $\frac{dln(y)}{dln(x)} = \beta_2$; Slope: $\frac{dy}{dx} = \frac{dln(y)}{dln(x)}\frac{y}{x} = \frac{\beta_2 y}{x}$

2. Reciprocal Model: $y = \beta_1 + \beta_2(1/x) \xrightarrow{t.d.wrt.x} \frac{dy}{dx} = -\beta_2 \frac{1}{x^2}$
   Elasticity: $\frac{dy}{dx}\frac{x}{y} = -\beta_2 \frac{1}{x^2}\frac{x}{y} = -\frac{\beta_2}{xy}$; Slope: $\frac{dy}{dx} = -\frac{\beta_2}{x^2}$

3. Linear-log Model: $y = \beta_1 + \beta_2 ln(x) \xrightarrow{t.d.wrt.x} \frac{dy}{dx} = \beta_2 \frac{dln(x)}{dx} = \frac{\beta_2}{x}$
   Elasticity: $\frac{dy}{dx}\frac{x}{y} = \frac{\beta_2}{x}\frac{x}{y} = \frac{\beta_2}{y}$; Slope: $\frac{dy}{dx} = \frac{\beta_2}{x}$

4. Log-linear Model: $ln(y) = \beta_1 + \beta_2 x \xrightarrow{t.d.wrt.x} \frac{dln(y)}{dx} = \beta_2$
   Elasticity: $\frac{dy}{dx}\frac{x}{y} = \frac{dln(y)x}{dx} = \beta_2 x$; Slope: $\frac{dy}{dx} = \beta_2 x \frac{y}{x} = \beta_2 y$

5. Liner Model: $y = \beta_1 + \beta_2 x \xrightarrow{t.d.wrt.x} \frac{dy}{dx} = \beta_2$
   Elasticity: $\frac{dy}{dx}\frac{x}{y} = \frac{\beta_2 x}{y}$; Slope: $\frac{dy}{dx} = \beta_2$

6. Quadratic Model: $y = \beta_1 + \beta_2 x^2 \xrightarrow{t.d.wrt.x} \frac{dy}{dx} = \beta_2 2x$
   Elasticity: $\frac{dy}{dx}\frac{x}{y} = \beta_2 2x\frac{x}{y} = \frac{2\beta_2 x^2}{y}$; Slope: $\frac{dy}{dx} = 2\beta_2 x$

7. Cubic Model: $y = \beta_1 + \beta_2 x^3 \xrightarrow{t.d.wrt.x} \frac{dy}{dx} = \beta_2 3x^2$
   Elasticity: $\frac{dy}{dx}\frac{x}{y} = \beta_2 3x^2 \frac{x}{y} = \frac{3\beta_2 x^3}{y}$; Slope: $3\beta_2 x^2$

PS 2: Poultry Demand

1. Residuals

```
*Model 1: Q = a_1 + a_2(1/P) + e
gen y1 = q //generate left-hand side variable y
gen x1 = 1/p //generate right-hand side regressor x
reg y1 x1 //do the regression
//Method 1:
predict Q1 //this gives us the fitted value of y1: Q1 = _b[_cons] + _b[x1]*x1
predict res1, residuals //this gives us the residual: res1 = y1-Q1
//Method 2:
gen Q1_test = _b[_cons] + _b[x1]*x1 //generate the fitted value of y1
gen res1_test = y1 - Q1_test //generate the residual by doing y1-Q1_test
```

2. Graph

```
*Generate the graph with scatter dots and fitted line
twoway (scatter q p) (function y=_b[_cons]+_b[x1]*(1/x), range(p)), ///
scheme(s1color) legend(label(1 "Real Consumptions") label(2 "Fitted Values")) ///
xtitle("Price of Chicken") ytitle("Quantity of Chicken") ///
title("Poultry Demand") subtitle("Reciprocal Model: Q=a1+a2(1/P)+e")
//the triple-slash "///" tells STATA continue working on the next line of code
//but it is only valid when you excuate the code in do-file
//it does not work if you excuate in the command window
//it might note work if you do not put a space in front of "///"
graph export m1.png, replace //export the current graph to the working directory
```

3. Elasticity and Correlation
Here the estimated model is: $\hat{Q} = -6.0244 + 48.3650P$ for the reciprocal model.
We know the Slope: $\frac{d\hat{Q}}{dP} = -\frac{48.3650}{P^2}$ and the Elasticity: $\frac{d\hat{Q}}{dP}\frac{P}{\hat{Q}} = -\frac{48.3650}{P\hat{Q}}$ by previous derivation.

```
*Calculate the derivative and the elasticity
scalar deri_1 = -_b[x1]/(1.31^2) //by the slope formula
scalar Q_hat_1 = _b[_cons]+_b[x1]*(1/1.31) //by the estimated model
scalar elas_1 = -_b[x1]/(1.31*Q_hat_1) //by the elasticity formula
display "the elasticity for reciprocal model is:" elas_1

*Model 4: lnQ = d_1 + d_2*P + e
gen y4 = ln(q)
gen x4 = p
reg y4 x4
predict Q4
predict res4, residuals
tabstat res4, statistics(variance) //tabstat compacts table of summary statistics
//the variance of the residual is 0.0250672
gen Q4_c = exp(_b[_cons]+_b[x4]*x4+(.0250672/2)) //by page 45 of lecture 4
twoway (scatter q p) (function y=exp(_b[_cons]+_b[x4]*x+(0.0250672/2)), range(p))
graph export m4.png, replace
//Method 1:
corr q Q4_c //generate the correlation
display r(rho)
scalar Rg_sqr = r(rho)^2
display Rg_sqr
//Method 2:
//use the definition in page 18-19 of lecture 4
//then you are able to calculate the correlation coefficient rho manually
scalar elas_4 = _b[x4]*1.31
display elas_4
```

4. Assumption Examination

```
*Using Diagnostic Residual Plots
//Exam assumption: homoskedasticity
scatter res1 x1
graph export res1_1.png, replace
//Exam assumption: no serial correlation
scatter res1 year, connect(l)
graph export res1_2.png, replace
//Exam assumption: normality
hist res1
graph export res1_3.png, replace
ssc install jb6 //install the user-written Jarque-Bera test package
jb6 res1 //show the Jarque-Bera test result
//Jarque-Bera test statistic follows chi-square distribution
//For the homework, you just follow the same p-value rule as we learned before
//The null hypothesis is the regression errors are normally distributed
//refer to page 169 of the textbook for more information
```